National Aeronautics and Space Administration
Goddard Earth Science
Data Information and Services Center (GES DISC)

README Document for the Nimbus-6 HIRS High Resolution Infrared Radiation Sounder Level 1 Calibrated Radiances for the Global Atmospheric Research Program Data Product

HIRSN6L1GARP

Last Revised 09/05/2014

Goddard Earth Sciences Data and Information Services Center (GES DISC)
http://disc.gsfc.nasa.gov
NASA Goddard Space Flight Center
Code 610.2
Greenbelt, MD 20771 USA

Prepared By:

James E. Johnson	09/05/2014
Name GES DISC GSFC Code 610.2	Date

Reviewed By:

Name	mm/dd/yyyy
Name GSFC Code xxx	Date

Name	mm/dd/yyyy
Name GSFC Code xxx	Date

Goddard Space Flight Center
Greenbelt, Maryland

Revision History

Revision Date	***Changes***	***Author***
09/05/2014	Original	James E. Johnson

Table of Contents

1. Introduction

This document provides basic information on using the Nimbus-6 High Resolution Infrared Radiation Sounder (HIRS) Level-1 data product.

1.1 Data Product Description

The High Resolution Infrared Radiation Sounder (HIRS) Level 1 data product contains calibrated measurements from 17 channels in the IR through visible part of the spectrum. The data were collected in support of the Global Atmospheric Research Program (GARP). The data files are in a binary record oriented structure. Each file typically contains one full orbit of data. Spatial coverage is global. The data are available from August 17, 1975 (day of year 229) through March 4, 1976 (day of year 64).

This product was previously available from the NASA National Space Science Data Center (NSSDC) included under the Merged HIRS/SCAMS Radn, Tmp & Humidity Sounding Data for GARP Data Sys Test product with the identifier ESAD-00017 (previously 75-052A-02B and 75-052A-10C).

1.1.1 The High Resolution Infrared Radiation Sounder

The High Resolution Infrared Radiation Sounder (HIRS) measured radiances primarily in five spectral regions: (1) seven channels near the 15-micrometer CO_2 absorption band to improve vertical temperature profile resolution, (2) two channels, 11.1 and 3.7 micrometers, for detecting clouds during day and night, (3) two channels, 8.2 and 6.7 micrometers, in the water vapor absorption band, (4) five channels in the 4.3-micrometer band for temperature sounding, and (5) one channel in the visible 0.69-micrometer region for cloud detection. The sounder consisted of a Cassegrain telescope, scanning mirror, dichromatic beam splitter, filter wheel, chopper, and associated electronics. The HIRS scanned the Earth's surface in a plane normal to the spacecraft's orbital path with a maximum scan angle of 30° to either side of nadir to provide data with a spatial resolution of 25 km. The HIRS instrument was similar to the Infrared Temperature Profile Radiometer (ITPR) flown on the previous Nimbus-5 satellite.

The swath width of the HIRS instrument is about 2400 km, with a spatial resolution of about 145 km at nadir to about 330 km at the scan edges. The scanning mechanism rotates once every 16 seconds. The Nimbus-6 HIRS was operational from June 15, 1975 until the instrument ceased functioning on May 31, 1976, due to jamming of the scan mechanism.

The principal investigator for the HIRS experiment was Mr. William L. Smith.

1.1.2 Nimbus-6 Overview

The Nimbus-6 satellite was successfully launched on June 12, 1975. The spacecraft included nine experiments: (1) a Temperature-Humidity Infrared Radiometer (THIR) for measuring day and night surface and cloud top temperatures, as well as the water vapor content of the upper atmosphere, (2) a High-Resolution Infrared Radiation Sounder (HIRS) for determining vertical temperature profiles, and the distribution of water vapor in the atmosphere, (3) the High Resolution Infrared Radiation Sounder (HIRS) for obtaining vertical profiles of temperature in the troposphere and abundances of liquid water and water vapor, (4) an Electrically Scanning Microwave Radiometer (ESMR) for determining liquid water content of clouds, the distribution and variation of sea ice cover, and land surfaces characteristics, (5) the Earth Radiation Budget (ERB) experiment for accurate measurements of radiation from the sun and earth, (6) a Limb Radiance Inversion Radiometer (LRIR) for determining the vertical distribution of temperature, ozone and water vapor in the stratosphere and lower mesosphere, (7) a Pressure Modulator Radiometer (PMR) for measuring the temperature structure of the upper stratosphere and mesosphere, (8) the Tropical Wind Energy Conversion and Reference Level Experiment (TWERLE) for determining upper atmospheric winds in the tropics, pressure gradients, and provide a reference level in-coordination with in-situ balloon measurements and (9) a Tracking and Data Relay Experiment (T&DRE) for demonstrating data communication from a low-orbiting spacecraft through a synchronous spacecraft to a ground telemetry station.

The orbit of the satellite can be characterized by the following:

- circular orbit at 1100 km
- inclination of 100 degrees
- period of an orbit is about 107.3 minutes
- orbits cross the equator at 26 degrees of longitude separation
- sun-synchronous

1.2 Algorithm Background

The Nimbus-6 HIRS data were generated from the spacecraft telemetry, attitude and orbital data. The data were originally processed on IBM 360/95 computers using a 32-bit architecture. The calibrated radiances from the several infrared and one visible channels can be used to derive vertical temperature profiles, detect clouds, and water vapor. Further information on the HIRS data processing and algorithms can be found in the Nimbus-6 Users' Guide Section 4.

1.3 Data Disclaimer

The data should be used care and one should first read the Nimbus-6 User's Guide, section 3 describing the HIRS experiment, as well as the <document>. Users should cite this data product in their research.

2. Data Organization

The Nimbus-6 High Resolution Infrared Radiation Sounder Level-1 data product spans the time period from August 17, 1975 to March 4, 1976. Each file typically contains about one orbit worth of data.

2.1 File Naming Convention

The data product files are named according to the following convention:

<Platform>-<Instrument>_<Date>_<Tape>.<Suffix>

where:

- Platform = name of the platform or satellite (always Nimbus6)
- Instrument = name of the instrument (always HIRS)
- Date = Data start date and time in UTC in format <YYYY>m<MMDD>t<hhmmss> where
    1. YYYY = 4 digit year (1975 or 1976)
    2. MM = 2 digit month (01-12)
    3. DD = 2 digit day of month (01-31)
    4. hh = 2 digit hour of day (01-23)
    5. mm = 2 digit minute of hour (01-59)
    6. dd = 2 digit second of minute (01-59)
- Tape = the id of the tape the file is from (e.g. DS1)
- Suffix = the file format (always TAP)

```
File name example: Nimbus6-HIRS_1975m0817t194751_DS882.TAP
```

2.2 File Format and Structure

The data are stored as they were originally written in IBM binary (big-endian) record oriented structured files, hereafter referred to as TAP. The files were written on the original 9-track tapes using a blocked FORTRAN format with a size that of 3600 byte size records (i.e. 900 IBM 4-byte words). On the original 9-track tapes, the HIRS Level-1 files were included with subsetted SCAMS and the merged HIRS and SCAMS retrieval data files.

This data product only includes the HIRS Level-1 data, the other data are archived as separate data products.

Figure 1: Original 9-Track Tape Structure of Level-1 Data



where N = number of files on tape

2.3 Key Science Data Fields

The primary science data fields in this data product are the calibrated radiances in units $mW/m^2/sr/cm$ for each of the 17 HIRS channels.

Figure 2: Example showing one orbit swath of IR calibrated radiances from HIRS channel 9 (H_2O band at 8.2 microns); darkest values = 0; lightest values = 55; units = $mW/m^2/sr/cm$.



3. Data Contents

The granularity of the data is typically one orbit or about 100 minutes of data per file.

3.1 Data Record

Note: Each data record is 900 words (3600 bytes). Table 3-6 in the Nimbus-6 User's Guide doesn't list the fields for words 886-900. The correct format is found in the Appendix B.

Table 3-1-1: Data fields within a record or major frame. All words are 4-byte integer types

Word	Field Name	Count	Units	Scale Factor
1	Time (GMT in Seconds)	1	-	1
2	Julian Day (1-366)	1	-	1
3	Year (75 or 76)	1	-	1
4-45	Data Quality Flag (0 = data acquired; 1 = no data acquired)	42	-	1
46-759	IR Calibrated Channel Data	17 x 42	$mW/m^2/sr/cm$	*See below
760-801	Latitude	42	degrees (+ N)	100
802-843	Longitude	42	degrees (+ E)	100
844-885	Zenith Angle	42	degrees	100
886	Line Number	1	-	1
887	Grid Number	1	-	1
888-900	Spare (Zero Filled)	13	-	-

Table 3-1-2: Scale Factors for IR Calibrated Channel Data

Channel	Scale Factor
1 - 10	100
11 - 16	10000
17	1

Table 3-1-3: The HIRS Channels

Channel Number	Central Frequency (cm^{-1})	Central Wavelength (μm)	Principal Absorbing Constituents	Level of PeakEnergy Contribution	Purpose of the Radiance Observation
1	668	15.0	CO_2	30 mb	Temperature Sounding. The 15 μm band channels provide better sensitivity to the temperature of relatively cold regions of the atmosphere than can be achieved with the 4.3 μm band channels. Radiances in Channels 5, 6 and 7 are also used to calculate the heights and amounts of cloud within the HIRS field of view.
2	679	14.7	CO_2	60 mb	
3	690	14.4	CO_2	100 mb	
4	702	14.2	CO_2/H_2O	250 mb	
5	716	14.0	CO_2/H_2O	500 mb	
6	733	13.6	CO_2	700 mb	
7	749	13.4	CO_2	900 mb	
8	900	11.0	Window	Surface	Surface Temperature and cloud detection
9	1224	8.2	H_2O	900 mb	Water Vapor Sounding. Provide water vapor corrections for CO_2 and window channels. The 6.7 μm channel is also used to detect thin cirrus cloud
10	1496	6.7	H_2O	400 mb	
11	2190	4.57	N_2O	950 mb	Temperature Sounding. The 4.3 μm band channels provide better sensitivity to the temperature of relatively warm regions of the atmosphere than can be achieved with the 15 μm band channels. Also, the short wavelength radiances are less sensitive to clouds than those for the 15 μm region.
12	2212	4.52	N_2O	850 mb	
13	2242	4.46	CO_2/N_2O	700 mb	
14	2275	4.40	CO_2/N_2O	600 mb	
15	2357	4.24	CO_2	5 mb	
16	2692	3.71	Window	Surface	Surface Temperature. Much less sensitive to clouds and H2O than 11 μm window. Used with 11 μm channel to detect cloud contamination and derive surface temperature under partly could sky conditions
17	14,443	0.69	Window	Cloud	Cloud Detection. Used during the day with 3.7 μm and 11 μm window channels to define clear fields of view and to specify any reflected solar contributions to the 3.7 μm channel

3.2 Metadata

The metadata are contained in a separate XML formatted file having the same name as the data file with .xml appended to it.

Table 3-2: Metadata attributes associated with the data file.

Name	Description
LongName	Long name of the data product.
ShortName	Short name of the data product.
VersionID	Product or collection version.
GranuleID	Granule identifier, i.e. the name of the file.
Format	File format of the data file.
CheckSumType	Type of checksum used.
CheckSumValue	The value of the calculated checksum.
SizeBytesDataGranule	Size of the file or granule in bytes.
InsertDateTime	Date and time when the granule was inserted into the archive. The format for date is YYYY-MM-DD and time is hh-mm-ss.
RangeBeginningDate	Begin date when the data was collected in YYYY-MM-DD format.
RangeBeginningTime	Begin time of the date when the data was collected in hh-mm-ss format.
RangeEndingDate	End date when the data was collected in YYYY-MM-DD format.
RangeEndingTime	End time of the date when the data was collected in hh-mm-ss format.
PlatformShortName	Short name or acronym of the platform or satellite
InstrumentShortName	Short name or acronym of the instrument
SensorShortName	Short name or acronym of the sensor
GPolygon: PointLatitude	Latitudes of the polygon (rectangle) points that represent the satellite coverage. Each point is identified by its latitude and longitude pair.
GPolygon: PointLongitude	Longitudes of the polygon (rectangle) points that represent the satellite coverage. Each point is identified by its latitude and longitude pair.
ElapsedMinTime	Duration in minutes of data collected during an orbit.

4. Reading the Data

The data are written in a proprietary binary record-oriented format. We provide the following sample FORTRAN code snippet for reading the data:

```
C-----------------------------------------------------------------------
C ^NAME: READ_HIRS
C
C ^DESCRIPTION:
C     This program opens and reads a Nimbus-6 HIRS level-1 data file
C     and prints the contents of the file to the screen. Data files
C     consist of records with 900 IBM 4-byte words (3600 bytes).
C     See the Nimbus-6 User's Guide Section 3 for the HIRS file
C     specification.
C
C ^MAJOR VARIABLES:
C     FNAME  - name of input file
C     DREC   - buffer for data records
C     IOS    - I/O status number
C
C ^NOTES:
C     Compile: gfortran -o READ_HIRS.EXE -frecord-marker=4 READ_HIRS.FOR
C
C ^ORGANIZATION: NASA/GSFC, Code 610.2
C
C ^AUTHOR:  James Johnson
C
C ^ADDRESS: james.johnson@nasa.gov
C
C ^CREATED: Sep. 10, 2014
C-----------------------------------------------------------------------

      CHARACTER          FNAME*1024    ! Name of input data file
      INTEGER*4          DREC(900)     ! Data record 900 IBM 4-byte words


C     Get the name of the input data file to read
      PRINT *, 'Enter the name of the input file:'
      READ (5,'(A)') FNAME

C     Open the specified input file using sequential access
      OPEN (UNIT=1, FILE=FNAME, STATUS='OLD', ACCESS='SEQUENTIAL',
     &      FORM='UNFORMATTED', IOSTAT=IOS, ERR=99)

C     Initialize N (block number)
      N=1

C     Loop through the file reading all blocks of data
      DO

        READ (1, IOSTAT=IOS, ERR=98, END=100) DREC
```

```
        PRINT '("***************")'
        PRINT '("RECORD =",X,I5)', N
        PRINT '("***************")'
        CALL PRDREC(DREC)

        N=N+1

      END DO

C     Close the input file
      CLOSE(1)
      GOTO 100

   98 PRINT '("ERROR: READ BLOCK ",I4,", IOSTAT: ",I5)', N, IOS
      GOTO 100
   99 PRINT '("ERROR: OPEN IOSTAT =",I5)', IOS

  100 STOP
      END


C--------------------------------------------------------------------------
C ^SUBROUTINE: PRDREC
C
C     This subroutine will print data records to the screen
C--------------------------------------------------------------------------

      SUBROUTINE PRDREC(DREC)

      INTEGER*4          DREC(900)     ! Data record 900 IBM 4-byte words
      INTEGER*4          TIME,         ! Time (GMT in seconds)
     &                   DAY,          ! Julian Day Number (Day of Year)
     &                   YEAR,         ! Year
     &                   QFLAGS(42),   ! Sounding Quality Flags (0=data;1=none)
     &                   IRDATA(17,42),! IR calibrated channel data
     &                   LAT(42),      ! Latitude (+ north)          ; scale=100
     &                   LON(42),      ! Longitude (+ east)          ; scale=100
     &                   ZENANG(42),   ! Zenith angle (+ northward)  ; scale=100
     &                   LINE,         ! Line number
     &                   GRID,         ! Grid number
     &                   SPARE(13)     ! Spare (zero filled)
      REAL*4             SCLFAC        ! Scale factor


      N = 1
      TIME = I4SWAP(DREC(N))
      PRINT '("TIME   =",X,I10)', TIME

      N = 2
      DAY = I4SWAP(DREC(N))
      PRINT '("DAY    =",X,I10)', DAY

      N = 3
      YEAR = I4SWAP(DREC(N))
      PRINT '("YEAR   =",X,I10)', YEAR
```

```
      N = 4-1
      DO 10 I=1,42
        QFLAGS(I) = I4SWAP(DREC(N+I))
   10 CONTINUE
      PRINT '("QFLAGS =",5(X,I10),/,(8X,5(X,I10)))', QFLAGS

      N = 46-1
      DO 21 K=1,17
        DO 20 I=1,42
          IRDATA(K,I) = I4SWAP(DREC(N+K+17*(I-1)))
   20   CONTINUE
        IF (K .LT. 11) THEN
          SCLFAC = 100.                ! IR Channels 1-10
        ELSE IF (K .LT. 17) THEN
          SCLFAC = 10000.              ! IR Channels 11-16
        ELSE
          SCLFAC = 1.                  ! IR Channel 17
        END IF
        PRINT '("IRCH",I2.2," =",5(X,G12.6),/,(8X,5(X,G12.6)))',
     &        K, (IRDATA(K,1:42) / SCLFAC)
   21 CONTINUE

      SCLFAC = 100.                    ! LAT, LON and ZENANG are scaled
      N = 760-1
      DO 30 I=1,42
        LAT(I) = I4SWAP(DREC(N+I))
   30 CONTINUE
      PRINT '("LAT    =",5(X,G12.6),/,(8X,5(X,G12.6)))', (LAT / SCLFAC)

      N = 802-1
      DO 40 I=1,42
        LON(I) = I4SWAP(DREC(N+I))
   40 CONTINUE
      PRINT '("LON    =",5(X,G12.6),/,(8X,5(X,G12.6)))', (LON / SCLFAC)

      N = 844-1
      DO 50 I=1,42
        ZENANG(I) = I4SWAP(DREC(N+I))
   50 CONTINUE
      PRINT '("ZENANG =",5(X,G12.6),/,(8X,5(X,G12.6)))', (ZENANG/SCLFAC)

      N = 886
      LINE = I4SWAP(DREC(N))
      PRINT '("LINE   =",X,I10)', LINE

      N = 887
      GRID = I4SWAP(DREC(N))
      PRINT '("GRID   =",X,I10)', GRID

      N = 888-1
      DO 60 I=1,13
        SPARE(I) = I4SWAP(DREC(N+I))
   60 CONTINUE
      PRINT '("SPARE  =",5(X,I10),/,(8X,5(X,I10)))', SPARE

      RETURN
      END
```

```
C-----------------------------------------------------------------------
C ^FUNCTION: I4SWAP
C
C     This subroutine will swap the bytes of a data element
C-----------------------------------------------------------------------

      FUNCTION I4SWAP(INPUT)

      INTEGER*4          IWORD          ! Input 4-byte word
      INTEGER*4          IDROW          ! Byte-swapped 4-byte word
      CHARACTER          DATBUF(4)      ! Input data buffer
      CHARACTER          SWPBUF(4)      ! Output swapped buffer
      EQUIVALENCE        (IWORD, DATBUF)
      EQUIVALENCE        (IDROW, SWPBUF)


      IWORD = INPUT
      DO 10 K=1,4
        SWPBUF(K) = DATBUF(4-K+1)
   10 CONTINUE
      I4SWAP = IDROW

      RETURN
      END
```

5. Data Services

5.1 Reverb

The GES DISC provides basic temporal and advanced (event) searches through the EOSDIS Reverb data search and download interface:

http://reverb.echo.nasa.gov

Reverb allows users the ability to search on keywords, spatial region, and time period on datasets archived and various data centers. It offers various download options that suit users with different preferences and different levels of technical skills. To search for the HIRS data enter GES_DISC_HIRSN6L1GARP_V001 into the keyword field.

5.2 FTP

The Nimbus data products are available for users to download directly using anonymous FTP:

ftp://acdisc.gsfc.nasa.gov/data/s4pa/Nimbus6_HIRS_SCAMS_Level1/HIRSN6L1GARP.001/

The data are organized in directories by year with subdirectories by day of year. README and other documentation are located under the doc directory.

6. More Information

6.1 Web Resources

For other Nimbus data products, please see the GES DISC's Nimbus heritage data web page at:
http://disc.gsfc.nasa.gov/nimbus/

To search for other related data, please visit NASA's Global Change Master Directory at:
http://gcmd.nasa.gov.

6.2 Point of Contact

Name: GES DISC Help Desk
URL: http://disc.gsfc.nasa.gov/
E-mail: gsfc-help-disc@lists.nasa.gov
Phone: 301-614-5224
Fax: 301-614-5228
Address: Goddard Earth Sciences Data and Information Services Center
Attn: Help Desk
Code 610.2
NASA Goddard Space Flight Center
Greenbelt, MD 20771, USA

6.3 References

W. L. Smith, et al, "The Nimbus-6 User's Guide - Section 3: The High Resolution Infrared Radiation Sounder (HIRS) Experiment", NASA Goddard Space Flight Center, Feb. 1975, Pages 37-58

Gary, J. P., "AOIPS Data Base Management Systems Support for GARP Data Sets, Appendix B", NASA Goddard Space Flight Center, Oct. 1977 (DSC_0585.pdf)

7. Acknowledgements

The Nimbus data recovery task at the GES DISC is funded by NASA's Earth Science Data and Information System program.

Acronyms

EOS: Earth Observing System

ESDIS: Earth Science and Data Information System

GES DISC: Goddard Earth Sciences Data and Information Services Center

GSFC: Goddard Space Flight Center

HIRS: High Resolution Infrared Radiation Sounder

L1: Level-1 Data

NASA: National Aeronautics and Space Administration

Reverb: ECHO's Next Generation Metadata and Service Discovery Tool

QA: Quality Assessment

UT: Universal Time